

NAG Library Routine Document

F07AQF (ZCGESV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07AQF (ZCGESV) computes the solution to a complex system of linear equations

$$AX = B,$$

where A is an n by n matrix and X and B are n by r matrices.

2 Specification

```
SUBROUTINE F07AQF (N, NRHS, A, LDA, IPIV, B, LDB, X, LDX, WORK, SWORK,      &
                  RWORK, ITER, INFO)
```

```
INTEGER                N, NRHS, LDA, IPIV(N), LDB, LDX, ITER, INFO
REAL (KIND=nag_wp)    RWORK(N)
COMPLEX (KIND=nag_wp) A(LDA,*), B(LDB,*), X(LDX,*), WORK(N*NRHS)
COMPLEX (KIND=nag_rp) SWORK(N*(N+NRHS))
```

The routine may be called by its LAPACK name *zcgesev*.

3 Description

F07AQF (ZCGESV) first attempts to factorize the matrix in single precision and use this factorization within an iterative refinement procedure to produce a solution with double precision accuracy. If the approach fails the method switches to a double precision factorization and solve.

The iterative refinement process is stopped if

$$\text{ITER} > \text{itermax},$$

where ITER is the number of iterations carried out thus far and *itermax* is the maximum number of iterations allowed, which is fixed at 30 iterations. The process is also stopped if for all right-hand sides we have

$$\|resid\| < \sqrt{N}\|x\|\|A\|\epsilon,$$

where $\|resid\|$ is the ∞ -norm of the residual, $\|x\|$ is the ∞ -norm of the solution, $\|A\|$ is the ∞ -operator-norm of the matrix A and ϵ is the *machine precision* returned by X02AJF.

The iterative refinement strategy used by F07AQF (ZCGESV) can be more efficient than the corresponding direct full precision algorithm. Since this strategy must perform iterative refinement on each right-hand side, any efficiency gains will reduce as the number of right-hand sides increases. Conversely, as the matrix size increases the cost of these iterative refinements become less significant relative to the cost of factorization. Thus, any efficiency gains will be greatest for a very small number of right-hand sides and for large matrix sizes. The cut-off values for the number of right-hand sides and matrix size, for which the iterative refinement strategy performs better, depends on the relative performance of the reduced and full precision factorization and back-substitution. For now, F07AQF (ZCGESV) always attempts the iterative refinement strategy first; you are advised to compare the performance of F07AQF (ZCGESV) with that of its full precision counterpart F07ANF (ZGESV) to determine whether this strategy is worthwhile for your particular problem dimensions.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Buttari A, Dongarra J, Langou J, Langou J, Luszczek P and Kurzak J (2007) Mixed precision iterative refinement techniques for the solution of dense linear systems *International Journal of High Performance Computing Applications*

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: N – INTEGER *Input*
On entry: n , the number of linear equations, i.e., the order of the matrix A .
Constraint: $N \geq 0$.
- 2: NRHS – INTEGER *Input*
On entry: r , the number of right-hand sides, i.e., the number of columns of the matrix B .
Constraint: $NRHS \geq 0$.
- 3: A(LDA,*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the n by n coefficient matrix A .
On exit: if iterative refinement has been successfully used (i.e., if $INFO = 0$ and $ITER \geq 0$), then A is unchanged. If double precision factorization has been used (when $INFO = 0$ and $ITER < 0$), A contains the factors L and U from the factorization $A = PLU$; the unit diagonal elements of L are not stored.
- 4: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F07AQF (ZCGESV) is called.
Constraint: $LDA \geq \max(1, N)$.
- 5: IPIV(N) – INTEGER array *Output*
On exit: if no constraints are violated, the pivot indices that define the permutation matrix P ; at the i th step row i of the matrix was interchanged with row $IPIV(i)$. $IPIV(i) = i$ indicates a row interchange was not required. IPIV corresponds either to the single precision factorization (if $INFO = 0$ and $ITER \geq 0$) or to the double precision factorization (if $INFO = 0$ and $ITER < 0$).
- 6: B(LDB,*) – COMPLEX (KIND=nag_wp) array *Input*
Note: the second dimension of the array B must be at least $\max(1, NRHS)$.
On entry: the n by r right-hand side matrix B .
- 7: LDB – INTEGER *Input*
On entry: the first dimension of the array B as declared in the (sub)program from which F07AQF (ZCGESV) is called.
Constraint: $LDB \geq \max(1, N)$.

- 8: X(LDX,*) – COMPLEX (KIND=nag_wp) array Output
Note: the second dimension of the array X must be at least $\max(1, \text{NRHS})$.
On exit: if INFO = 0, the n by r solution matrix X.
- 9: LDX – INTEGER Input
On entry: the first dimension of the array X as declared in the (sub)program from which F07AQF (ZCGESV) is called.
Constraint: $\text{LDX} \geq \max(1, N)$.
- 10: WORK(N * NRHS) – COMPLEX (KIND=nag_wp) array Workspace
- 11: SWORK(N * (N + NRHS)) – COMPLEX (KIND=nag_rp) array Workspace
Note: this array is utilized in the reduced precision computation, consequently its type nag_rp reflects this usage.
- 12: RWORK(N) – REAL (KIND=nag_wp) array Workspace
- 13: ITER – INTEGER Output
On exit: if ITER > 0, iterative refinement has been successfully used and ITER is the number of iterations carried out.
 If ITER < 0, iterative refinement has failed for one of the reasons given below and double precision factorization has been carried out instead.
 ITER = -1
 Taking into account machine parameters, and the values of N and NRHS, it is not worth working in single precision.
 ITER = -2
 Overflow of an entry occurred when moving from double to single precision.
 ITER = -3
 An intermediate single precision factorization failed.
 ITER = -31
 The maximum permitted number of iterations was exceeded.
- 14: INFO – INTEGER Output
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element $\langle value \rangle$ of the diagonal is exactly zero. The factorization has been completed, but the factor U is exactly singular, so the solution could not be computed.

7 Accuracy

The computed solution for a single right-hand side, \hat{x} , satisfies the equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and ϵ is the *machine precision*. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1}$$

where $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$, the condition number of A with respect to the solution of the linear equations. See Section 4.4 of Anderson *et al.* (1999) for further details.

8 Parallelism and Performance

F07AQF (ZCGESV) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F07AQF (ZCGESV) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The real analogue of this routine is F07ACF (DSGESV).

10 Example

This example solves the equations

$$Ax = b,$$

where A is the general matrix

$$A = \begin{pmatrix} -1.34 + 2.55i & 0.28 + 3.17i & -6.39 - 2.20i & 0.72 - 0.92i \\ -0.17 - 1.41i & 3.31 - 0.15i & -0.15 + 1.34i & 1.29 + 1.38i \\ -3.29 - 2.39i & -1.91 + 4.42i & -0.14 - 1.35i & 1.72 + 1.35i \\ 2.41 + 0.39i & -0.56 + 1.47i & -0.83 - 0.69i & -1.96 + 0.67i \end{pmatrix} \quad \text{and} \quad b = \begin{pmatrix} 26.26 + 51.78i \\ 6.43 - 8.68i \\ -5.75 + 25.31i \\ 1.16 + 2.57i \end{pmatrix}.$$

10.1 Program Text

Program f07aqfe

```
!      F07AQF Example Program Text
!
!      Mark 25 Release. NAG Copyright 2014.
!
!      .. Use Statements ..
!      Use nag_library, Only: nag_rp, nag_wp, zcgesev
!      .. Implicit None Statement ..
!      Implicit None
!      .. Parameters ..
!      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
!      Integer                     :: i, info, iter, j, lda, ldb, ldx, n, r
!      .. Local Arrays ..
```

```

Complex (Kind=nag_wp), Allocatable :: a(:,:), b(:,:), work(:), x(:,:)
Complex (Kind=nag_wp), Allocatable :: swork(:)
Real (Kind=nag_wp), Allocatable :: rwork(:)
Integer, Allocatable :: ipiv(:)
! .. Executable Statements ..
Write (nout,*) 'F07AQF Example Program Results'
Write (nout,*)
! Skip heading in data file
Read (nin,*)
Read (nin,*) n, r
lda = n
ldb = n
ldx = n
Allocate (a(lda,n),b(ldb,r),work(n*r),x(ldx,r),swork(n*(n+ &
r)),ipiv(n),rwork(n))

! Read A and B from data file

Read (nin,*)(a(i,1:n),i=1,n)
Read (nin,*)(b(i,1:r),i=1,n)

! Solve the equations Ax = b for x

! The NAG name equivalent of zcgsv is f07aqf
Call zcgsv(n,r,a,lda,ipiv,b,ldb,x,ldx,work,swork,rwork,iter,info)

If (info==0) Then

! Print solution

Write (nout,*) 'Solution'
Write (nout,99999)((x(i,r),j=1,r),i=1,n)

! Print pivot indices

Write (nout,*)
Write (nout,*) 'Pivot indices'
Write (nout,99998) ipiv(1:n)

Else
Write (nout,99997) 'The (' , info, ', ', info, ')', &
' element of the factor U is zero'
End If

99999 Format ((3X,4(' (',F7.4,',',F7.4,')':)))
99998 Format (1X,7I11)
99997 Format (1X,A,I3,A,I3,A,A)
End Program f07aqfe

```

10.2 Program Data

F07AQF Example Program Data

```

4      1                                     :Value of N, R

(-1.34, 2.55) ( 0.28, 3.17) (-6.39,-2.20) ( 0.72,-0.92)
(-0.17,-1.41) ( 3.31,-0.15) (-0.15, 1.34) ( 1.29, 1.38)
(-3.29,-2.39) (-1.91, 4.42) (-0.14,-1.35) ( 1.72, 1.35)
( 2.41, 0.39) (-0.56, 1.47) (-0.83,-0.69) (-1.96, 0.67) :End of matrix A

(26.26,51.78) ( 6.43,-8.68) (-5.75,25.31) ( 1.16, 2.57) :End of vector b

```

10.3 Program Results

F07AQF Example Program Results

Solution

(1.0000, 1.0000) (2.0000,-3.0000) (-4.0000,-5.0000) (0.0000, 6.0000)

Pivot indices

3 2 3 4
